// File: contracts-verify/interfaces/IKUMAUniswapStaking.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity 0.8.17;

interface IKUMAUniswapStaking {
    event AccessControllerSet(address newAccessController);
    event StablecoinSet(address stablecoin);
    event WrappedKIBTSet(address KIBT);
    event NewRouterSet(address router);
    event NewUniswapFeeSet(uint24 newFee);
    event Deposit(address stableSwapPool, address user, uint256 stablecoinAmount, uint256 KIBTBought);
    event DepositWrappedKIBT(address indexed user, uint256 wrappedKIBTAmount);
    event Withdraw(address indexed pool, address indexed user, uint256 wrappedKIBTAmount, uint256 stablecoinBought);
    event WithdrawWrappedKIBT(address indexed user, uint256 wrappedKIBTAmount);
    event SharesTransferred(address indexed sender, address indexed receiver, uint256 sharesTransferred);

    function deposit(uint256 stablecoinAmount, uint256 minKIBTOut) external returns (uint256);
    function depositWrappedKIBT(uint256 wrappedKIBTDepositAmount) external;
    function withdraw(uint256 wrappedKIBTWithdrawAmount, uint256 minStablecoinOut)
        external
        returns (uint256 stablecoinBought);
    function withdrawWrappedKIBT(uint256 wrappedKIBTWithdrawAmount) external;
    function transferShares(uint256 shareTransferAmount, address receiver) external;
    function setRouter(address newUniswapRouter) external;
    function setUniswapFee(uint24 newFee) external;
    function getWithdrawableWrappedKIBT(address user) external view returns (uint256);
    function getShares(address user) external view returns (uint256);
    function getTotalShares() external view returns (uint256);
    function getRouter() external view returns (address);
}

// File: contracts-verify/interfaces/uniswap-core/interfaces/callback/IUniswapV3SwapCallback.sol



// SPDX-License-Identifier: GPL-2.0-or-later
pragma solidity >=0.5.0;

/// @title Callback for IUniswapV3PoolActions#swap
/// @notice Any contract that calls IUniswapV3PoolActions#swap must implement this interface
interface IUniswapV3SwapCallback {
    /// @notice Called to `msg.sender` after executing a swap via IUniswapV3Pool#swap.
    /// @dev In the implementation you must pay the pool tokens owed for the swap.
    /// The caller of this method must be checked to be a UniswapV3Pool deployed by the canonical UniswapV3Factory.
    /// amount0Delta and amount1Delta can both be 0 if no tokens were swapped.
    /// @param amount0Delta The amount of token0 that was sent (negative) or must be received (positive) by the pool by
    /// the end of the swap. If positive, the callback must send that amount of token0 to the pool.
    /// @param amount1Delta The amount of token1 that was sent (negative) or must be received (positive) by the pool by
    /// the end of the swap. If positive, the callback must send that amount of token1 to the pool.
    /// @param data Any data passed through by the caller via the IUniswapV3PoolActions#swap call
    function uniswapV3SwapCallback(int256 amount0Delta, int256 amount1Delta, bytes calldata data) external;
}

// File: contracts-verify/interfaces/uniswap-core/interfaces/IUniswapV3Factory.sol



// SPDX-License-Identifier: GPL-2.0-or-later
pragma solidity >=0.5.0;

/// @title The interface for the Uniswap V3 Factory
/// @notice The Uniswap V3 Factory facilitates creation of Uniswap V3 pools and control over the protocol fees
interface IUniswapV3Factory {
    /// @notice Emitted when the owner of the factory is changed
    /// @param oldOwner The owner before the owner was changed
    /// @param newOwner The owner after the owner was changed
    event OwnerChanged(address indexed oldOwner, address indexed newOwner);

    /// @notice Emitted when a pool is created
    /// @param token0 The first token of the pool by address sort order
    /// @param token1 The second token of the pool by address sort order
    /// @param fee The fee collected upon every swap in the pool, denominated in hundredths of a bip
    /// @param tickSpacing The minimum number of ticks between initialized ticks
    /// @param pool The address of the created pool
    event PoolCreated(
        address indexed token0, address indexed token1, uint24 indexed fee, int24 tickSpacing, address pool
    );

    /// @notice Emitted when a new fee amount is enabled for pool creation via the factory
    /// @param fee The enabled fee, denominated in hundredths of a bip
    /// @param tickSpacing The minimum number of ticks between initialized ticks for pools created with the given fee
    event FeeAmountEnabled(uint24 indexed fee, int24 indexed tickSpacing);

    /// @notice Returns the current owner of the factory
    /// @dev Can be changed by the current owner via setOwner
    /// @return The address of the factory owner
    function owner() external view returns (address);

    /// @notice Returns the tick spacing for a given fee amount, if enabled, or 0 if not enabled
    /// @dev A fee amount can never be removed, so this value should be hard coded or cached in the calling context
    /// @param fee The enabled fee, denominated in hundredths of a bip. Returns 0 in case of unenabled fee
    /// @return The tick spacing
    function feeAmountTickSpacing(uint24 fee) external view returns (int24);

    /// @notice Returns the pool address for a given pair of tokens and a fee, or address 0 if it does not exist
    /// @dev tokenA and tokenB may be passed in either token0/token1 or token1/token0 order
    /// @param tokenA The contract address of either token0 or token1
    /// @param tokenB The contract address of the other token
    /// @param fee The fee collected upon every swap in the pool, denominated in hundredths of a bip
    /// @return pool The pool address
    function getPool(address tokenA, address tokenB, uint24 fee) external view returns (address pool);

    /// @notice Creates a pool for the given two tokens and fee
    /// @param tokenA One of the two tokens in the desired pool
    /// @param tokenB The other of the two tokens in the desired pool
    /// @param fee The desired fee for the pool
    /// @dev tokenA and tokenB may be passed in either order: token0/token1 or token1/token0. tickSpacing is retrieved
    /// from the fee. The call will revert if the pool already exists, the fee is invalid, or the token arguments
    /// are invalid.
    /// @return pool The address of the newly created pool
    function createPool(address tokenA, address tokenB, uint24 fee) external returns (address pool);

    /// @notice Updates the owner of the factory
    /// @dev Must be called by the current owner
    /// @param _owner The new owner of the factory
    function setOwner(address _owner) external;

    /// @notice Enables a fee amount with the given tickSpacing
    /// @dev Fee amounts may never be removed once enabled
    /// @param fee The fee amount to enable, denominated in hundredths of a bip (i.e. 1e-6)
    /// @param tickSpacing The spacing between ticks to be enforced for all pools created with the given fee amount
    function enableFeeAmount(uint24 fee, int24 tickSpacing) external;
}

// File: contracts-verify/interfaces/uniswap-periphery/interfaces/ISwapRouter.sol



// SPDX-License-Identifier: GPL-2.0-or-later
pragma solidity >=0.7.5;
pragma abicoder v2;

import "../../uniswap-core/interfaces/callback/IUniswapV3SwapCallback.sol";

/// @title Router token swapping functionality
/// @notice Functions for swapping tokens via Uniswap V3
interface ISwapRouter is IUniswapV3SwapCallback {
    struct ExactInputSingleParams {
        address tokenIn;
        address tokenOut;
        uint24 fee;
        address recipient;
        uint256 deadline;
        uint256 amountIn;
        uint256 amountOutMinimum;
        uint160 sqrtPriceLimitX96;
    }

    /// @notice Swaps `amountIn` of one token for as much as possible of another token
    /// @param params The parameters necessary for the swap, encoded as `ExactInputSingleParams` in calldata
    /// @return amountOut The amount of the received token
    function exactInputSingle(ExactInputSingleParams calldata params) external payable returns (uint256 amountOut);

    struct ExactInputParams {
        bytes path;
        address recipient;
        uint256 deadline;
        uint256 amountIn;
        uint256 amountOutMinimum;
    }

    /// @notice Swaps `amountIn` of one token for as much as possible of another along the specified path
    /// @param params The parameters necessary for the multi-hop swap, encoded as `ExactInputParams` in calldata
    /// @return amountOut The amount of the received token
    function exactInput(ExactInputParams calldata params) external payable returns (uint256 amountOut);

    struct ExactOutputSingleParams {
        address tokenIn;
        address tokenOut;
        uint24 fee;
        address recipient;
        uint256 deadline;
        uint256 amountOut;
        uint256 amountInMaximum;
        uint160 sqrtPriceLimitX96;
    }

    /// @notice Swaps as little as possible of one token for `amountOut` of another token
    /// @param params The parameters necessary for the swap, encoded as `ExactOutputSingleParams` in calldata
    /// @return amountIn The amount of the input token
    function exactOutputSingle(ExactOutputSingleParams calldata params) external payable returns (uint256 amountIn);

    struct ExactOutputParams {
        bytes path;
        address recipient;
        uint256 deadline;
        uint256 amountOut;
        uint256 amountInMaximum;
    }

    /// @notice Swaps as little as possible of one token for `amountOut` of another along the specified path (reversed)
    /// @param params The parameters necessary for the multi-hop swap, encoded as `ExactOutputParams` in calldata
    /// @return amountIn The amount of the input token
    function exactOutput(ExactOutputParams calldata params) external payable returns (uint256 amountIn);

    function factory() external view returns (address);
}

// File: contracts-verify/KUMAUniswapStaking.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity 0.8.17;

import {IERC20} from "lib/kuma-protocol/lib/openzeppelin-contracts/contracts/interfaces/IERC20.sol";
import {SafeERC20} from "lib/kuma-protocol/lib/openzeppelin-contracts/contracts/token/ERC20/utils/SafeERC20.sol";
import {IKUMAUniswapStaking} from "./interfaces/IKUMAUniswapStaking.sol";
import {IAccessControl} from "lib/kuma-protocol/lib/openzeppelin-contracts/contracts/access/IAccessControl.sol";
import {Roles} from "./libraries/Roles.sol";
import {KUMAStakingErrors} from "./libraries/Errors.sol";
import {ReentrancyGuard} from "lib/kuma-protocol/lib/openzeppelin-contracts/contracts/security/ReentrancyGuard.sol";
import {IWrappedRebaseToken} from "lib/wrapped-rebase-tokens/src/interfaces/IWrappedRebaseToken.sol";
import {ISwapRouter} from "./interfaces/uniswap-periphery/interfaces/ISwapRouter.sol";
import {IUniswapV3Factory} from "./interfaces/uniswap-core/interfaces/IUniswapV3Factory.sol";

contract KUMAUniswapStaking is IKUMAUniswapStaking, ReentrancyGuard {
    using SafeERC20 for IERC20;

    IAccessControl public immutable accessController;
    IERC20 public immutable stablecoin; // The stablecoin that can be swapped to staked wKIBT
    IWrappedRebaseToken public immutable wKIBT; // The wrapped KIBT that can be staked and swapped for stablecoin

    ISwapRouter private _uniswapRouter;
    uint24 private _uniswapFee;
    uint256 private _totalShares;
    mapping(address => uint256) private _shares; // Mapping of indiviual users to their wrappedRebaseToken balances

    /**
     * @notice Checks that msg.sender has a specific `role` access
     * @param role Bytes32 role from the Roles library
     */
    modifier onlyRole(bytes32 role) {
        if (!accessController.hasRole(role, msg.sender)) {
            revert KUMAStakingErrors.ACCESS_CONTROL_ACCOUNT_IS_MISSING_ROLE(msg.sender, role);
        }
        _;
    }

    /**
     * @param _accessController KUMA AccessController
     * @param _stablecoin Stablecoin that can be swapped to staked wKIBT
     * @param _wKIBT Wrapped KIBT that is escrowed in this contract
     * @param uniswapRouter Address of the Uniswap V3 router that can be used to for swaps
     * @param uniswapFee Fee of the Uniswap pool that is used for swaps
     */
    constructor(
        IAccessControl _accessController,
        IERC20 _stablecoin,
        IWrappedRebaseToken _wKIBT,
        ISwapRouter uniswapRouter,
        uint24 uniswapFee
    ) {
        if (
            address(_accessController) == address(0) || address(_stablecoin) == address(0)
                || address(_wKIBT) == address(0) || address(uniswapRouter) == address(0)
        ) {
            revert KUMAStakingErrors.CANNOT_SET_TO_ADDRESS_ZERO();
        }

        accessController = _accessController;
        stablecoin = _stablecoin;
        wKIBT = _wKIBT;
        _uniswapRouter = uniswapRouter;
        _uniswapFee = uniswapFee;

        emit AccessControllerSet(address(_accessController));
        emit StablecoinSet(address(_stablecoin));
        emit WrappedKIBTSet(address(_wKIBT));
        emit NewRouterSet(address(uniswapRouter));
        emit NewUniswapFeeSet(uniswapFee);
    }

    /**
     * @notice Swap stablecoin for wKIBT and holds the wKIBT into this contract
     * @notice  Deposited wKIBT - swap fees can be later withdrawn using withdraw
     * @dev Must approve this contract to spend stablecoin for stablecoinAmount before this can be called
     * @param stablecoinInAmount Amount of stablecoin to exchange for wKIBT, in stablecoin decimals
     * @param minWrappedKIBTOut Minimum amount of wKIBT to receive from the exchange, in wKIBT decimals
     * @return Actual amount of wKIBT received in swap that will now accrue in this contract
     */
    function deposit(uint256 stablecoinInAmount, uint256 minWrappedKIBTOut) external nonReentrant returns (uint256) {
        if (stablecoinInAmount == 0) {
            revert KUMAStakingErrors.CANNOT_DEPOSIT_ZERO_TOKENS();
        }
        // Cache variables for gas savings
        IERC20 stablecoin_ = stablecoin;
        address wKIBT_ = address(wKIBT);
        uint24 uniswapFee_ = _uniswapFee;
        ISwapRouter uniswapRouter_ = _uniswapRouter;

        stablecoin_.safeTransferFrom(msg.sender, address(this), stablecoinInAmount);
        stablecoin_.safeIncreaseAllowance(address(uniswapRouter_), stablecoinInAmount);

        uint256 _wKIBTBought = uniswapRouter_.exactInputSingle(
            ISwapRouter.ExactInputSingleParams({
                tokenIn: address(stablecoin_),
                tokenOut: wKIBT_,
                fee: uniswapFee_,
                recipient: address(this),
                deadline: block.timestamp,
                amountIn: stablecoinInAmount,
                amountOutMinimum: minWrappedKIBTOut,
                sqrtPriceLimitX96: 0 // Bound by minKIBTOut so we don't need to set a limit
            })
        );

        _shares[msg.sender] += _wKIBTBought;
        _totalShares += _wKIBTBought;

        emit Deposit(
            IUniswapV3Factory(uniswapRouter_.factory()).getPool(address(stablecoin_), wKIBT_, uniswapFee_),
            msg.sender,
            stablecoinInAmount,
            _wKIBTBought
        );
        return _wKIBTBought;
    }

    /**
     * @notice Deposit wrapped rebase tokens to this staking contract directly
     * @notice Similar to deposit, but without swapping from the stablecoin
     * @dev Increments a user's share depending on the amount of deposited wrapped rebase tokens
     * @dev Requires approval of the wrapped rebase token for the depositAmount
     * @param wrappedKIBTDepositAmount Amount of wrapped rebase tokens to deposit, in wrapped rebase token decimals
     */
    function depositWrappedKIBT(uint256 wrappedKIBTDepositAmount) external {
        if (wrappedKIBTDepositAmount == 0) {
            revert KUMAStakingErrors.CANNOT_DEPOSIT_ZERO_TOKENS();
        }
        _shares[msg.sender] += wrappedKIBTDepositAmount;
        _totalShares += wrappedKIBTDepositAmount;
        IERC20(address(wKIBT)).safeTransferFrom(msg.sender, address(this), wrappedKIBTDepositAmount);
        emit DepositWrappedKIBT(msg.sender, wrappedKIBTDepositAmount);
    }

    /**
     * @notice Swap staked wKIBT for stablecoin and send received stablecoin to user
     * @param wrappedKIBTWithdrawAmount Amount of wKIBT to withdraw and swap for stablecoin, in wKIBT decimals
     * @param minStablecoinOut Minimum amount of stablecoin to receive from the exchange, in stablecoin decimals
     * @return stablecoinBought Actual amount of stablecoin received from swap
     */
    function withdraw(uint256 wrappedKIBTWithdrawAmount, uint256 minStablecoinOut)
        external
        nonReentrant
        returns (uint256 stablecoinBought)
    {
        if (wrappedKIBTWithdrawAmount == 0) {
            revert KUMAStakingErrors.CANNOT_WITHDRAW_ZERO_TOKENS();
        }
        uint256 userShares_ = _shares[msg.sender];
        if (userShares_ < wrappedKIBTWithdrawAmount) {
            revert KUMAStakingErrors.REQUESTED_WITHDRAW_AMOUNT_HIGHER_THAN_USER_BALANCE(
                msg.sender, wrappedKIBTWithdrawAmount, userShares_
            );
        }

        // Cache variables for gas savings
        address wKIBT_ = address(wKIBT);
        address stablecoin_ = address(stablecoin);
        ISwapRouter uniswapRouter_ = _uniswapRouter;
        uint24 uniswapFee_ = _uniswapFee;

        IERC20(wKIBT_).safeIncreaseAllowance(address(uniswapRouter_), wrappedKIBTWithdrawAmount);

        _shares[msg.sender] = userShares_ - wrappedKIBTWithdrawAmount;
        _totalShares -= wrappedKIBTWithdrawAmount;

        stablecoinBought = uniswapRouter_.exactInputSingle(
            ISwapRouter.ExactInputSingleParams({
                tokenIn: address(wKIBT_),
                tokenOut: stablecoin_,
                fee: uniswapFee_,
                recipient: msg.sender,
                deadline: block.timestamp,
                amountIn: wrappedKIBTWithdrawAmount,
                amountOutMinimum: minStablecoinOut,
                sqrtPriceLimitX96: 0 // Bound by minKIBTOut so we don't need to set a limit
            })
        );

        emit Withdraw(
            address(IUniswapV3Factory(uniswapRouter_.factory()).getPool(stablecoin_, address(wKIBT_), uniswapFee_)),
            msg.sender,
            wrappedKIBTWithdrawAmount,
            stablecoinBought
        );
    }

    /**
     * @notice Withdraw a user's wrapped rebase tokens without swapping on the Curve Pool
     * @notice Useful if there is low liquidity in the curve pool or if the user wants to keep wrapped rebase tokens
     * @param wrappedKIBTWithdrawAmount Amount of wKIBT to withdraw, in wKIBT decimals
     */
    function withdrawWrappedKIBT(uint256 wrappedKIBTWithdrawAmount) external {
        if (wrappedKIBTWithdrawAmount == 0) {
            revert KUMAStakingErrors.CANNOT_WITHDRAW_ZERO_TOKENS();
        }
        uint256 userShares_ = _shares[msg.sender];
        if (userShares_ < wrappedKIBTWithdrawAmount) {
            revert KUMAStakingErrors.REQUESTED_WITHDRAW_AMOUNT_HIGHER_THAN_USER_BALANCE(
                msg.sender, wrappedKIBTWithdrawAmount, userShares_
            );
        }

        _shares[msg.sender] = userShares_ - wrappedKIBTWithdrawAmount;
        _totalShares -= wrappedKIBTWithdrawAmount;
        IERC20(address(wKIBT)).safeTransfer(msg.sender, wrappedKIBTWithdrawAmount);
        emit WithdrawWrappedKIBT(msg.sender, wrappedKIBTWithdrawAmount);
    }

    /**
     * @notice Transfers the stake of the sender to another user
     * @param shareTransferAmount Amount of shares to transfer, in 27 decimals
     * @param receiver Address of the receiver
     */
    function transferShares(uint256 shareTransferAmount, address receiver) external {
        if (shareTransferAmount == 0) {
            revert KUMAStakingErrors.CANNOT_TRANSFER_ZERO_SHARES();
        }
        if (receiver == address(0)) {
            revert KUMAStakingErrors.CANNOT_TRANSFER_TO_ADDRESS_ZERO();
        }
        if (receiver == msg.sender) {
            revert KUMAStakingErrors.CANNOT_TRANSFER_TO_SELF();
        }
        if (shareTransferAmount > _shares[msg.sender]) {
            revert KUMAStakingErrors.REQUESTED_TRANSFER_AMOUNT_HIGHER_THAN_USER_BALANCE(
                msg.sender, shareTransferAmount, _shares[msg.sender]
            );
        }

        _shares[msg.sender] -= shareTransferAmount;
        _shares[receiver] += shareTransferAmount;
        emit SharesTransferred(msg.sender, receiver, shareTransferAmount);
    }

    /**
     * @notice Sets the address of the Uniswap v3 router
     * @dev Can only be called by the `KUMA_MANAGER_ROLE`.
     * @param newUniswapRouter New Uniswap v3 router.
     */
    function setRouter(address newUniswapRouter) external onlyRole(Roles.KUMA_MANAGER_ROLE) {
        if (address(newUniswapRouter) == address(0)) {
            revert KUMAStakingErrors.CANNOT_SET_TO_ADDRESS_ZERO();
        }
        _uniswapRouter = ISwapRouter(newUniswapRouter);
        emit NewRouterSet(address(newUniswapRouter));
    }

    /**
     * @notice Sets the Uniswap v3 fee
     * @dev Can only be called by the `KUMA_MANAGER_ROLE`.
     * @param newFee New Uniswap v3 fee.
     */
    function setUniswapFee(uint24 newFee) external onlyRole(Roles.KUMA_MANAGER_ROLE) {
        _uniswapFee = newFee;
        emit NewUniswapFeeSet(newFee);
    }

    /**
     * @notice Return the amount of wrapped KIBT that can be withdrawn by the user in 18 decimals
     * @param user Address of the user to query
     */
    function getWithdrawableWrappedKIBT(address user) external view returns (uint256) {
        return _shares[user];
    }

    /**
     * @notice Return the amount of shares for a user; equivalent to the amount of wrapped KIBT staked for the user
     * @param user Address of the user to query
     */
    function getShares(address user) external view returns (uint256) {
        return _shares[user];
    }

    /**
     * @notice Return the total amount of shares in this contract; equivalent to the total amount of wrapped KIBT staked
     */
    function getTotalShares() external view returns (uint256) {
        return _totalShares;
    }

    /**
     * @notice Return uniswap fee for the pool
     */
    function getUniswapFee() external view returns (uint24) {
        return _uniswapFee;
    }

    /**
     * @notice Return the address of the Uniswap v3 router
     */
    function getRouter() external view returns (address) {
        return address(_uniswapRouter);
    }
}

// File: contracts-verify/libraries/Errors.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;

library KUMAStakingErrors {
    error ACCESS_CONTROL_ACCOUNT_IS_MISSING_ROLE(address account, bytes32 role);
    error CANNOT_SET_TO_ADDRESS_ZERO();
    error CANNOT_TRANSFER_TO_ADDRESS_ZERO();
    error CANNOT_DEPOSIT_ZERO_TOKENS();
    error CANNOT_TRANSFER_ZERO_SHARES();
    error CANNOT_TRANSFER_TO_SELF();
    error CANNOT_WITHDRAW_ZERO_TOKENS();
    error REQUESTED_WITHDRAW_AMOUNT_HIGHER_THAN_USER_BALANCE(address user, uint256 requestedAmount, uint256 balance);
    error REQUESTED_TRANSFER_AMOUNT_HIGHER_THAN_USER_BALANCE(
        address user, uint256 requestedShareTransferAmount, uint256 shareBalance
    );
    error INVALID_POOL_INDICIES(uint256 stablecoinPoolIndex, uint256 KIBTPoolIndex);
}

// File: contracts-verify/libraries/Roles.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity 0.8.17;

library Roles {
    bytes32 public constant KUMA_POOL_ACCESS_ROLE = keccak256("KUMA_POOL_ACCESS_ROLE");
    bytes32 public constant KUMA_MANAGER_ROLE = keccak256("KUMA_MANAGER_ROLE");
    bytes32 public constant MCAG_MINT_ROLE = keccak256("MCAG_MINT_ROLE");
    bytes32 public constant MCAG_BURN_ROLE = keccak256("MCAG_BURN_ROLE");
    bytes32 public constant MCAG_BLACKLIST_ROLE = keccak256("MCAG_BLACKLIST_ROLE");
}

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: lib/kuma-protocol/lib/openzeppelin-contracts/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: lib/wrapped-rebase-tokens/src/interfaces/IWrappedRebaseToken.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;

import {IERC20} from "lib/kuma-protocol/lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol";

interface IWrappedRebaseToken is IERC20 {
    function wrap(uint256 depositAmount) external;

    function unWrap(uint256 amount) external;
}